



# SASによる迷路の自動生成

松沢 享

(株式会社インターサイエンティフィックリサーチ)

## Creating a maze in SAS

Akira Matsuzawa

Inter Scientific Research Co.,Ltd

## 要旨：

SAS言語のデータハンドリング機能は日々進化しており、これまでSASでは面倒とされていた処理が容易に行えるようになった。

本発表では、その進化の一例として、従来のデータステップと新機能を用いて迷路を自動生成するSASプログラムを作成したので紹介する。

キーワード：棒倒し法, 穴掘り法, 配列, Report Writing Interface, Hashオブジェクト



## 迷路を生成するアルゴリズム

迷路を生成するアルゴリズムは以下が有名である。

- ✓ 棒倒し法
- ✓ 穴掘り法
- ✓ 壁伸ばし法

今回は入門編として、一番簡単なアルゴリズムかつプログラミングが容易な「棒倒し法」を用いて解説する。




## 棒倒し法とは？

①




縦と横が奇数のマスを用意

②




準備として、外周のマスと、縦×横が奇数のマスを塗りつぶす

③




1段目の奇数マスからスタート。上下左右いずれかのマスを選び塗りつぶす

④




上記は右を塗りつぶした例。次の奇数マスに進むが、左が既に塞がっているので、上下右のいずれかのマスを選び塗りつぶす

⑤



こうして1段目が終わったら、2段目も同様に選んで塗りつぶしていく。但し2段目以降は下左右から選ぶ。

⑥



こうして最後まで行ったら、スタートとゴール地点を書き加えて迷路完成！



## SASプログラムの概要

- ✓ 棒倒し法 …… 配列を利用
- ✓ 迷路の描画 …… RWI (Report Writing Interface) を利用
- ✓ 迷路生成のポイント …… 初期値の設定

```
%let n = 37;      * 迷路の1辺のマス数 (仕組み上奇数を指定する);  
%let init = 1234; * 乱数のシード値;
```

 初期値を変えると迷路の構造を変えることが出来る



## 棒倒し法のプログラム

```
%let n = 37; * 迷路の1辺のマス数 (仕組み上奇数を指定する);
%let init = 1234; * 乱数のシード値 ;

data DT1;
  *** 全マス目分の配列を定義 ;
  array AR(&n, &n);

  *** 最初から壁にする部分を2に設定 ;
  do y=1 to &n;
  do x=1 to &n;
    if x in (1,&n) or y in (1,&n) or mod(x*y,2)=1 then AR(x,y)=2;
    else AR(x,y)=1;
  end;
end;

  *** 乱数を生成して壁の位置を決める 壁の位置 1:上、2:下、3:右、4:左 ;
  call streaminit(&init);
  do y=3 to &n-2 by 2;
  do x=3 to &n-2 by 2;
```

```
*** 1段目は上下左右の壁の乱数を生成 ;
if y=3 then do;
  * 左のマスが壁だったら上下右のみの壁の乱数を生成 ;
  if AR(x-1,y)=2 then RAND = ceil(rand('uniform')*3);
  else RAND = ceil(rand('uniform')*4);
end;

*** 2段目以降は下左右の壁の乱数を生成 ;
if y^=3 then do;
  * 左のマスが壁だったら下右のみの壁の乱数を生成 ;
  if AR(x-1,y)=2 then RAND = ceil(rand('uniform')*2)+1;
  else RAND = ceil(rand('uniform')*3)+1;
end;

if RAND=1 then AR(x,y-1)=2; * 上 ;
if RAND=2 then AR(x,y+1)=2; * 下 ;
if RAND=3 then AR(x+1,y)=2; * 右 ;
if RAND=4 then AR(x-1,y)=2; * 左 ;
end;
end;
run;
```




## 迷路を描画するプログラム

```
data _NULL_;
  length VAR1 $5.;
  set DT1;
  array AR(&n, &n);
  dcl odsout ob();
  ob.table_start();
  do y=1 to &n;
    ob.row_start();
    do x=1 to &n;
      *** 壁と床を描画 ;
      VAR1="";
      if (x=2 and y=2) or (x=&n-1 and y=&n-1) then VAR1="☆";
      ob.format_cell(data:VAR1, style_attr: "background=" ||
                    choosec(AR(x,y),"white","black") || " height=4mm width=4mm");
    end;
    ob.row_end();
  end;
  ob.table_end();
run;
```






## 結果



もう少し作りこんでみる。

- ✓ アルゴリズムを穴掘り法に変更・・・ Hashオブジェクトを利用
- ✓ 見た目をより迷路っぽくする


結果



プログラムはSASプレミアムラウンジで公開しています

本発表によってSASプログラミングの進化と楽しさを感じて頂けると幸いです。

もう少し作りこんでみる。(おまけ)



```
* 設定した初期値 ;  
%let n = 79;  
%let init = 123;
```

## 参考文献

- ・柳井 政和(2014)「プログラマのためのコードパズル ～JavaScriptで挑むコードゴルフとアルゴリズム」技術評論社
- ・SAS9.4 Output Delivery System: Advanced Topics Second Edition  
(<http://support.sas.com/documentation/cdl/en/odsadvug/67385/PDF/default/odsadvug.pdf>)
- ・SAS9.4 コンポーネントオブジェクト: リファレンス 第二版  
(<http://www.sas.com/offices/asiapacific/japan/service/help/pdf/v94/lecompobjref.pdf>)
- ・松沢 享(2015)「RWIによる柔軟なレポート作成 一覧表やカレンダーを作る例に」SASユーザー総会2015
- ・森岡 裕, 神田 悟志(2014)「SASハッシュオブジェクトを利用して医薬品開発に使用するプログラムを効率化する  
ー有害事象と併用薬、臨床検査値と途中変更のある施設基準値のマッチングからSASプログラムコードの分析まで」SASユーザー総会2014